

## SAS Macros for the Area Under the Curve (AUC) Calculation of Continuous Glucose Monitoring (CGM) Data

Xiao Ran Han, The Chinese University of Hong Kong, China  
Ka Chun Chong, The Chinese University of Hong Kong, China

### ABSTRACT

Continuous glucose monitoring (CGM) system, which provides an alternative approach measuring real-time glucose levels, have been increasingly adopted in the daily management of diabetes in the past decade. In addition to monitoring the glucose levels, CGM system also provides the glycemic variability, which associates with diabetes complications. In this paper, we developed SAS macros to perform general CGM data analysis. The use of SAS macros was demonstrated with the example of CGM data from a diabetic trial.

### INTRODUCTION

Diabetes mellitus (DM) is a chronic disease due to the pancreas incapable of producing enough insulin, or the cells of the body fail to respond to the insulin it produced adequately. As a global issue, around 425 million people are living with DM and this number will increase to 693 million in 2045 [1]. Subjects with DM are often present with excessive urinary frequency (polyuria), thirst (polydipsia), constant hunger (polyphagia), fatigue and weight loss. Complications of DM may occur if lack of good management. Acute complications may result in diabetic ketoacidosis (DKA), nonketotic hyperosmolar coma, loss of consciousness, or death. Long-term complications include microvascular damage (such as retinopathy, nephropathy, and nerves) and macrovascular damage (such as heart attacks and strokes) [2].

Continuous glucose monitoring (CGM) system is a device consisting of three parts: a sensor which measures interstitial glucose concentrations, a transmitter which sends data to the display device and a receiver which displays real-time glycemic information. A CGM system allows users to gain their real-time glucose concentration every five minutes (figure 1). Besides glucose levels, data collected by CGM system also provide additional information about the glycemic variability throughout the day, such as the amplitude, trend, duration in blood glucose concentrations [3, 4]. Previous studies have demonstrated that glycemic variability is an independent risk factor for diabetes complications, especially for cardiovascular disease (CVD) [5]. In this paper, we establish SAS macros to perform CGM data analysis. By calling SAS macros, glycemic parameters such as the area under the curve (AUC) can be calculated.




Figure 1. 24-hour CGM data for one example subject

## CGM DATASET

Dataset (CGMS.sas7bdat) used to fit macros in this paper is from a clinical trial about diabetes. Variables included in this dataset are listed in the table below:

Variable Name	Variable Label	Data Type	Variable Format
SUBJID	Subject identifier	CHARACTER	\$6.
CGMSDTM	Date and time of CGMS measurement	DATE	Datetime19.
CGMSDT	Date of CGMS measurement	DATE	DDMMYY10.
CGMSTM	Time of CGMS measurement	DATE	Time8.
MEALTP	Meal type	CHARACTER	\$10.
BGVAL	Blood glucose value (mmol/L)	NUM	3.1

**Table 1. Dataset templates for SAS macros**

## MACRO FOR POST-PRANDIAL AUC CALCULATION OVER A GIVEN TIME DURATION

After a meal, the glucose in plasma will increase which stimulates the pancreas to release insulin for lowering the blood glucose concentration and meanwhile inhibit hepatic glucose production. The post-prandial AUC of glucose over a certain time duration could reflect the function of the glycemic control system. Therefore, we developed a SAS macro for AUC calculation, following are the parameters used in this macro:

TIMEDURV – Time duration for AUC calculation

AUCOUTV, OUTDAT1 – Value and dataset of output AUC for each meal

AUCOUTDV, OUTDAT2 – Value and dataset of output AUC for each day

TMCRIT – Time criteria for AUC calculation

```
%macro AUCPP(timedurv, aucoutv, aucoutdv, tmcrit, outdat1, outdat2);

  proc sql;
    create table CGMSTMSP as
      select SUBJID, CGMSDTM as MEALDTM label="Start Date/time of meal",
      MEALTP, BGVAL,

      /* Number the meal type for following sort step */
      case
        when MEALTP ="BREAKFAST" then 1
        when MEALTP="LUNCH" then 2
        when MEALTP ="DINNER" then 3
        else .
      end as MEALTPN label="Meal Type (NUM)",

      /* Create the timestamp (in second) for AUC calculation */
      CGMSDTM+&timedurv as TIMESTAMP format=datetime19.
    from CGMS
    where not missing(MEALTP)
    order by SUBJID, MEALDTM, MEALTP;
  quit;

  proc sql;
    /* Select the glucose level, the meal type number and the corresponding
    dates and times for AUC calculation */
    create table aucdata as
      select A.SUBJID, A.CGMSDTM, A.CGMSDT, A.BGVAL, B.MEALTPN, B.MEALDTM,
      B.TIMESTAMP
    from CGMS as A, CGMSTMSP as B
```

SAS Macros for the Area Under the Curve (AUC) Calculation of Continuous Glucose Monitoring (CGM) Data, continued

```

        where A.SUBJID=B.SUBJID and B.MEALDTM<=A.CGMSDTM<=B.TIMESTAMP and not
missing(B.TIMESTAMP)
        order by A.SUBJID, A.CGMSDT, B.MEALTPN, B.MEALDTM, A.CGMSDTM;
quit;

/* Get the interval of glucose measurement time in minutes and calculate AUC
for each hour */
data auc;
    set aucdata;
    TIMEDIF=dif(CGMSDTM)/60; /* in minutes */
    AUC=((BGVAL+lag(BGVAL))* TIMEDIF)/120; /* in hours */
run;

/* Imput the zero point of AUC */
data auci;
    set auc;
    by SUBJID CGMSDT MEALTPN MEALDTM TIMESTAMP;
    if first.MEALTPN then do;
        AUC=0;
        TIMEDIF=5;
    end;
run;

proc sql;
    /* Calculate the total AUC per meal */
    create table aucpm as
    select SUBJID, CGMSDT, MEALTPN, MEALDTM, TIMESTAMP,
        sum(AUC) as AUCPM format=6.2,

        /* Time after meals to ensure enough time length for AUC calculation */
        sum(TIMEDIF) as DIFPM
    from auci
    group by SUBJID, CGMSDT, MEALTPN, MEALDTM, TIMESTAMP;
quit;

/* To filter out the AUC that overlap with the following one */
proc sort data=aucpm;
    by SUBJID descending MEALDTM;
run;
data aucpmf;
    set aucpm;
    LMESTDTM=lag(MEALDTM);
    if TIMESTAMP > LMESTDTM then overlaf1=1;
    else overlaf1=0;
run;

/* To prevent the first subject record is flagged */
data aucpmfi;
    set aucpmf;
    by SUBJID;
    if first.SUBJID then overlaf1=0;
run;

proc sql;
    /* Get AUC per meal satisfying the criterial of valid time length and
flag */
    create table &outdat1 as
    select SUBJID, CGMSDT, MEALTPN,
        AUCPM as &aucoutv format=6.2 label="Post-prandial AUC (per
meal) "
    from aucpmfi

```

```

/* having DIFPM >= time criteria for enough time length for AUC
calculation as some meals may not have enough timestamps */
where DIFPM >= &tmcrit and overlaf1=0 and MEALTPN in (1,2,3)
order by SUBJID, CGMSDT, MEALTPN;

/* Calculate the mean AUC per day */
create table &outdat2 as
select SUBJID, CGMSDT,
      mean(aucoutv) as &aucoutdv format=6.2 label="Post-prandial AUC
(per day) "
      from &outdat1
      group by SUBJID, CGMSDT;
quit;

%mend AUCPP;

```

With this SAS macro, the post-prandial AUC for each meal over a given time duration (&timedurv), as well as the mean AUC per day, can be calculated. By specifying the time criteria (&tmcrit), only the value of AUC satisfying the requirement will get output. Here's an example of two output tables of post-prandial AUC, one calculated for each meal the other each day, when &timedurv is set to 3600 and &tmcrit set to 60:

Subject ID	Date of CGMS	Meal type (NUM)	Post-prandial AUC (per meal)
Test01	13/04/2015	1	3.71
Test01	13/04/2015	2	5.80
Test01	13/04/2015	3	4.59
Test01	14/04/2015	1	6.82
.....	.....	.....	.....
Test03	15/04/2015	3	9.83

**Table 2. Macro output per meal**

Subject ID	Date of CGMS	Post-prandial AUC (per day)
Test01	13/04/2015	4.70
Test01	14/04/2015	6.58
Test01	15/04/2015	5.77
Test02	13/04/2015	5.26
.....	.....	.....
Test03	15/04/2015	8.93

**Table 3. Macro output per day**

### MACRO FOR CALCULATING THE POST-PRANDIAL AUC ABOVE A SPECIFIED LEVEL OVER A GIVEN TIME DURATION

Clinical studies have demonstrated that subjects with hyperglycemia after meals have more risk of developing diabetes. The management focusing on post-prandial hyperglycemia is helpful for preventing or delaying the onset of diabetic complications. To measure the AUC of blood glucose above a targeted range we developed a SAS macro, following are the parameters used in this macro:

- TIMEDURV – Time duration for AUC calculation
- GPVAR – Group variable for grouping AUC output
- BGLEVEL - The specified glucose level for AUC calculation
- AUCOUTV, OUTDAT – Value and dataset of output AUC

```

%macro AUCHYPER (timedurv, gpvar, bglevel, aucoutv, outdat);
  proc sql;
    create table CGMSTMSP as
      select SUBJID, CGMSDTM as MEALDTM label="Start Date/time of meal",
MEALTP, BGVAL,

      /* Number the meal type for following sort step */
      case
        when MEALTP ="BREAKFAST" then 1
        when MEALTP ="LUNCH" then 2
        when MEALTP ="DINNER" then 3
        else .
      end as MEALTPN label="Meal Type (NUM)",

      /* Create the timestamp (in second) for AUC calculation */
      CGMSDTM+&timedurv as TIMESTAMP format=datetime19.
    from CGMS
    where not missing(MEALTP)
    order by SUBJID, MEALDTM, MEALTP;
  quit;

  proc sql;
    /* Select the glucose level, the meal type number and the corresponding
dates and times for AUC calculation */
    create table aucdata as
      select A.SUBJID, A.CGMSDTM, A.CGMSDT, A.BGVAL, B.MEALTPN, B.MEALDTM,
B.TIMESTAMP
    from CGMS as A, CGMSTMSP as B
    where A.SUBJID=B.SUBJID and B.MEALDTM<=A.CGMSDTM<=B.TIMESTAMP and not
missing(B.TIMESTAMP)
    order by A.SUBJID, A.CGMSDT, B.MEALTPN, B.MEALDTM, A.CGMSDTM;
  quit;

  /* Count of grouping variable */
  %let varno=%sysfunc(countw(&gpvar));
  %let outgpvar=;
  %do i=1 %to &varno;

    /* Scan each variable */
    %let eachvar=%qscan(&gpvar,&i,%str( ));
    %if &i=&varno %then %let comma=;
    %else %let comma=,;
    %let outgpvar=&outgpvar&eachvar&comma;
  %end;

  /* Create lags and leads for trapezoidal calculation */
  proc expand data= aucdata out=auc2411 method=none;
    by SUBJID &gpvar;
    convert BGVAL = BGVALLAG / transformout=(lag 1);
    convert BGVAL;
    convert BGVAL = BGVALLEA / transformout=(lead 1);
  run;

  data auc180pp;
    set auc2411;
    /* Calculate only for glucose levels larger than the specified value (in
hour scale i.e. 1/12) */
    if BGVAL>&bglevel then do;
      /* Situation for area calculation of left and right triangles */
      if BGVALLAG<=&bglevel and BGVALLEA<=&bglevel then do;
        /* TMLAG=(BGVAL-bglevel)*(1/12)*((BGVAL-bglevel)/(BGVAL-
BGVALLAG)) */

```

```

        /* TMLEAD=(BGVAL-bglevel)*(1/12)*((BGVAL-bglevel)/(BGVAL-
BGVALLEA)) */
        AUCPERH=(BGVAL-&bglevel)*(1/24)*((BGVAL-&bglevel)/(BGVAL-
BGVALLAG)+(BGVAL-&bglevel)*(1/24)*((BGVAL-&bglevel)/(BGVAL-BGVALLEA)));
        end;
        /* Situation for area calculation of left triangle and right
rectangle */
        else if BGVALLAG<=&bglevel and BGVALLEA>&bglevel then do;
        /* TMLAG=(BGVAL-bglevel)*(1/12)*((BGVAL-bglevel)/(BGVAL-
BGVALLAG)) */
        /* TMLEAD=((BGVAL-bglevel)+(BGVALLEA-bglevel))*(5/2) */
        AUCPERH=(BGVAL-&bglevel)*(1/24)*((BGVAL-&bglevel)/(BGVAL-
BGVALLAG))+((BGVAL-&bglevel)+(BGVALLEA-&bglevel))*((1/12)/2);
        end;
        /* Situation for area calculation of whole rectangle */
        else if BGVALLAG>&bglevel and BGVALLEA>&bglevel then do;
        /* TMLAG=0 */
        /* TMLEAD=((BGVAL-bglevel)+(BGVALLEA-bglevel))*((1/12)/2)
*/
        AUCPERH=((BGVAL-&bglevel)+(BGVALLEA-&bglevel))*((1/12)/2);
        end;
        /* Situation for area calculation of left rectangle and right
triangle */
        else if BGVALLAG>&bglevel and BGVALLEA<=&bglevel then do;
        /* TMLAG=0 */
        /* TMLEAD=(BGVAL-bglevel)*(1/12)*((BGVAL-bglevel)/(BGVAL-
BGVALLEA)) */
        AUCPERH=(BGVAL-&bglevel)*(1/24)*((BGVAL-&bglevel)/(BGVAL-
BGVALLEA));
        end;
        end;
        /* Zero for glucose levels <= bglevel */
        else AUCPERH=0;
run;

proc sql;
        /* Calculate the total AUC */
        create table &outdat as
        select SUBJID, &outgpvar,
                sum(AUCPERH) as &aucoutv format=6.2 label="Post-prandial AUC above
the specified level"
        from auc180pp
        group by SUBJID, &outgpvar;
quit;

%mend AUCHYPER;

```

With this SAS macro, we could calculate the post-prandial AUC above a specified level (&bglevel) over a given time duration (&timedurv) by specifying the group variable (&gpvar) to group the output. Here's an example of the output table of post-prandial AUC when &timedurv is set to 7200, &gpvar set to CGMSDT and &bglevel set to 8:

Subject ID	Date of CGMS	Post-prandial AUC above the specified level
Test01	13/04/2015	0.00
Test01	14/04/2015	0.68
Test01	15/04/2015	0.00
Test02	13/04/2015	0.00
.....	.....	.....
Test03	15/04/2015	7.70

**Table 4. Macro output per day**

## CONCLUSION

CGM system as a technology, providing both the absolute glucose magnitude and the trend patterns in real-time, has been widely applied recently. Unlike intermittent values, glycemic parameters derived from frequently monitoring data allow researchers gain more insights into the function of the glucose-insulin system. In this paper, we proposed two SAS macros, one to calculate the post-prandial AUC of blood glucose over a given time duration, the other to calculate the AUC larger than a targeted level, and demonstrated the example dataset and output. In summary, the SAS macros assist researchers to conduct CGM data analysis conveniently.

## REFERENCES

1. Atlas, I. D. (2017). In., 6th edn. Brussels, Belgium: International Diabetes Federation; 2013. URL: <https://www.idf.org/component/attachments/attachments.html>.
2. World Health Organization. (2016). World health statistics 2016: monitoring health for the SDGs sustainable development goals. World Health Organization.
3. Soliman, A., DeSanctis, V., Yassin, M., Elalaily, R., & Eldarsy, N. E. (2014). Continuous glucose monitoring system and new era of early diagnosis of diabetes in high risk groups. *Indian journal of endocrinology and metabolism*, 18(3), 274.
4. Klonoff, D. C. (2005). Continuous glucose monitoring: roadmap for 21st century diabetes therapy. *Diabetes care*, 28(5), 1231-1239.
5. Danne, T., Nimri, R., Battelino, T., Bergenstal, R. M., Close, K. L., DeVries, J. H., ... & Beck, R. (2017). International consensus on use of continuous glucose monitoring. *Diabetes Care*, 40(12), 1631-1640.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Marc Chong

Enterprise: Centre for Clinical Research and Biostatistics, The Chinese University of Hong Kong

Address: Rm502, JC School of Public Health and Primary Care, The Chinese University of Hong Kong

Work Phone: +852 2252 8715

Fax: +852 2646 7297

E-mail: [marc@cuhk.edu.hk](mailto:marc@cuhk.edu.hk)

Web: [www.ccrb.cuhk.edu.hk](http://www.ccrb.cuhk.edu.hk)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.